Лабораторная работа №2
ОПЕРАЦИИ С ЯЧЕЙКАМИ И РАБОЧИМИ ЛИСТАМИ 
MS EXCEL В ПРОГРАММАХ НА VBA
Цель работы – Освоение разработки программ на VBA для обработки данных, размещенных в рабочих листах табличного процессора MS Excel.
2.1	Основные способы ссылок на ячейки рабочего листа Excel
Простейшие способы ссылки на значения отдельных ячеек – Range(ячейка).Value или Cells(строка, столбец).Value.
Пример 2.1 – Программа считывает число из ячейки A1, возводит его в квадрат и выводит результат в ячейку A3.
Sub primer2_1()
x = Range("A1").Value
y = x ^ 2
Range("A3").Value = y
End Sub
Это же можно реализовать и многими другими способами, например:
Sub primer2_1()
x = Cells(1, 1).Value
y = x ^ 2
Cells(3, 1).Value = y
End Sub
или
Sub primer2_1()
Cells(3, 1).Value = (Cells(1, 1).Value) ^ 2
End Sub
В качестве адресов ячеек могут использоваться не только конкретные значения, но и переменные.
Пример 2.2 – Требуется, чтобы программа считывала число из ячейки, указанной пользователем, возводила это число в квадрат и выводила результат в другую ячейку, также указанную пользователем.
Sub primer2_2()
adres_dannyh = InputBox("Укажите ячейку с исходными данными")
adres_rez = InputBox("Укажите ячейку для вывода результата")
x = Range(adres_dannyh).Value
y = x ^ 2
Range(adres_rez).Value = y
End Sub
Например, если требуется, чтобы число для возведения в квадрат считывалось из ячейки A5, а результат выводился в ячейку B7, то в ответ на запрос "Укажите ячейку с исходными данными" необходимо ввести A5, а на запрос "Укажите ячейку для вывода результата" – B7. Буквы A и B, конечно, должны быть латинскими.
Можно также ссылаться на ячейки, отсчитывая их не от левого верхнего угла рабочего листа (т.е. не от ячейки A1), а от некоторой заданной ячейки. Например, ссылка Range(“E7”).Cells(3,2).Value означает ссылку на значение ячейки F9, так как, если отсчитывать ячейки от E7, то ячейка F9 находится в третьей строке и втором столбце от заданной ячейки. Эту же ссылку можно указать как Cells(7,5).Cells(3,2).Value. Конечно, вместо конкретных номеров строк и столбцов могут указываться переменные. Например, ссылка Cells(7,5).Cells(i,j).Value – этот ссылка на ячейку, расположенную в i-й строке и j-м столбце относительно ячейки E7.
Ссылки на ячейки могут также указываться относительно ячейки, выделенной с помощью мыши. Например, ссылка Selection.Cells(3,2).Value означает ссылку на значение ячейки, находящейся в третьей строке и втором столбце от выделенной ячейки. Если при этом выделено несколько ячеек, то ссылка определяется относительно левого верхнего угла выделенного диапазона.
Если требуется указать рабочий лист, на котором находится ячейка, то перед ссылкой на нее (т.е. перед словом Range, Cells или Selection) указывается слово Worksheets("имя_листа").
Пример 2.3 – На рабочем листе Лист1 в столбец введено десять чисел. Курсор находится в первой из этих десяти ячеек. Требуется скопировать на другой рабочий лист (имя листа – Лист5) числа, превышающие 25. Числа должны копироваться в столбец E, начиная с ячейки E2. Между числами не должно быть пустых ячеек. Другими словами, если из десяти заданных чисел только три числа превысят 25, то они должны быть выведены в ячейки E2, E3 и E4 на рабочем листе Лист5.
Sub primer2_3()
j = 0
For i = 1 To 10
x = Selection.Cells(i, 1).Value
If x > 25 Then
j = j + 1
Worksheets("Лист5").Range("E2").Cells(j, 1) = x
End If
Next i
End Sub
Здесь переменная j – номер строки на рабочем листе Лист5 (начиная с ячейки E2), куда будет копироваться число из исходного набора данных. Сначала эта переменная принимается равной нулю. Цикл For i = 1 To 10 используется для перебора десяти исходных ячеек. В операторе x = Selection.Cells(i, 1).Value значение ячейки из i-й строки текущего столбца присваивается переменной x. При этом не требуется указывать, что ячейка находится на рабочем листе Лист1, так как этот рабочий лист – текущий. Если значение x превышает 25, то переменная j увеличивается на единицу, и переменная x выводится в соответствующую ячейку рабочего листа Лист5.
2.2	Основные способы ссылок на диапазоны ячеек 
рабочего листа Excel. Определение размеров диапазона ячеек
Основные способы ссылок на диапазоны ячеек в Excel следующие:
· ссылка на прямоугольный диапазон в форме Range(левый_верхний_угол, правый_нижний_угол), например, Range(“A2:E8”) или Range(Cells(2,1),Cells(8,5));
· ссылка на диапазон ячеек, выделенный на рабочем листе с помощью мыши: Selection;
· ссылка на прямоугольный диапазон ячеек, заполненный данными (например, числами) и отделенный от других данных хотя бы одной свободной строкой и столбцом: ячейка.CurrentRegion, где ячейка – ссылка на любую ячейку в заполненном диапазоне. Примеры таких обозначений: Range(“E2”).CurrentRegion – ссылка на заполненный диапазон, содержащий ячейку E2; Cells(2,5).CurrentRegion – то же самое (так как ячейка во второй строке и пятом столбце – это и есть ячейка E2); Selection.CurrentRegion – ссылка на заполненный диапазон, содержащий выделенную ячейку.
Во многих случаях удобно, указав диапазон ячеек одним из этих способов, ссылаться на отдельные ячейки в этом диапазоне в форме Cells(строка, столбец). Номера строк и столбцов в этом случае отсчитываются не от левого верхнего угла рабочего листа (т.е. не от ячейки A1), а от левого верхнего угла заданного диапазона. Например, ссылка Range(“A2:E8”).Cells(1,1).Value означает ссылку на значение ячейки A2, а Range(“A2:E8”).Cells(2,4).Value – ссылка на значение ячейки D3 (так как D3 – ячейка во второй строке, третьем столбце диапазона A2:E8).
Часто требуется определять количество строк и столбцов в диапазоне ячеек. Приведем несколько примеров:
· диапазон, заданный явно:
x = Range(“A2:E8”).Rows.Count
y = Range(“A2:E8”).Columns.Count
Здесь переменная x получит значение 7 (т.е. количество строк в диапазоне A2:E8), а переменная y – значение 5 (количество столбцов в диапазоне A2:E8);
· выделенный диапазон:
x = Selection.Rows.Count
y = Selection.Columns.Count
Здесь переменная x получит значение, равное количеству строк в диапазоне, выделенном с помощью мыши. Переменная y получит значение, равное количеству столбцов в этом диапазоне;
· диапазон, заполненный данными, где A2 – одна из ячеек этого диапазона:
x = Range(“A2”).CurrentRegion.Rows.Count
y = Range(“A2”).CurrentRegion.Columns.Count
Здесь переменные x и y получат значения, равные, соответственно, количеству строк и столбцов в заполненном диапазоне. Это же можно указать следующим образом:
x = Cells(2,1).CurrentRegion.Rows.Count
y = Cells(2,1).CurrentRegion.Columns.Count
Перед ссылками на диапазоны ячеек может указываться конструкция Worksheets("имя_листа"), если требуется указать рабочий лист, на котором располагается диапазон.
Диапазоны ячеек можно присваивать переменным. Для этого используется следующий оператор: Set переменная = диапазон. Пример:
Set d = Range(“A2:E8”)
После этого вместо ссылок на диапазон A2:E8 достаточно будет указать переменную d. Например, чтобы определить количество строк и столбцов в этом диапазоне, можно указать:
x = d.Rows.Count
y = d.Columns.Count
Чтобы присвоить переменной z значение ячейки B7, требуется указать: z = d.Cells(6,2).
Важно понимать, что в операторе Set переменная = диапазон переменной присваивается именно объект (диапазон ячеек), а не конкретное значение какой-либо ячейки.
Пример 2.4 – Пусть в условиях примера 2.3 количество чисел, введенных в столбец на рабочем листе Лист1, неизвестно. Требуется скопировать числа, превышающие 25, на рабочий лист Лист5 в столбец E, начиная с ячейки E2.
Sub primer2_4()
Set d=Selection.CurrentRegion
m=d.Rows.Count
j = 0
For i = 1 To m
x = Selection.Cells(i, 1).Value
If x > 25 Then
j = j + 1
Worksheets("Лист5").Range("E2").Cells(j, 1) = x
End If
Next i
End Sub
Чтобы эта программа выполнялась правильно, при ее запуске должна быть выделена одна из ячеек (необязательно первая) в столбце с числами на рабочем листе Лист1. В операторе Set d=Selection.CurrentRegion переменная d связывается с заполненным диапазоном ячеек, содержащим выделенную ячейку. Затем в операторе m=d.Rows.Count переменной m присваивается количество строк в заполненном диапазоне, т.е. количество чисел в столбце. Далее программа выполняется точно так же, как в примере 2.3.
2.3	Основные операции с рабочими листами MS Excel
Рассмотрим две важные операции с рабочими листами – создание нового листа и смену текущего листа – на следующем примере.
Пример 2.5 – Пусть в условиях примера 2.3 рабочий лист Лист5, на который требуется выводить отобранные числа, еще не существует. Программа в этом случае будет следующей.
Sub primer2_5()
Set NewSheet = Worksheets.Add
NewSheet.Name = "Лист5"
Worksheets("Лист1").Activate
j = 0
…См. пример 2.3

End Sub
Здесь в операторе Set NewSheet = Worksheets.Add создается новый рабочий лист. При этом он становится текущим. В операторе NewSheet.Name = "Лист5" ему присваивается имя Лист5. В операторе Worksheets("Лист1").Activate текущим становится рабочий лист Лист1. Последующий текст программы такой же, как в примере 2.3.
2.4	Примеры решения задач с использованием диапазонов ячеек
Пример 2.6 – Некоторый прямоугольный диапазон ячеек на рабочем листе Excel заполнен числами. Требуется вычислить средние значения каждой строки этого диапазона ячеек и вывести их справа от диапазона. Если, например, числами заполнен диапазон B1:E5, то требуется сначала вычислить среднее значение ячеек B1:E1 и вывести его в ячейку F1, затем вычислить среднее значение ячеек B2:E2 и вывести его в ячейку F2, и т.д.
Рассмотрим несколько случаев решения этой задачи в зависимости от того, как задан диапазон ячеек, для которого требуется вычислить средние значения строк.
a) Вычисления выполняются для диапазона ячеек B1:E5.
Sub primer2_6a_1()
Set d = Range("B1:E5")
m = d.Rows.Count
n = d.Columns.Count
For i = 1 To m
sum = 0
For j = 1 To n
sum = sum + d.Cells(i, j).Value
Next j
srednee = sum / n
d.Cells(i, n + 1).Value = srednee
Next i
End Sub
Здесь переменной d присваивается диапазон ячеек B1:E5. Переменные m и n получают значения, равные количеству строк и столбцов этого диапазона. Затем вычисляется среднее по каждой строке этого диапазона. Важно обратить внимание, что ссылка Cells(i, j) – это ссылка на ячейку с номером строки i и номером столбца j, причем номера строк и столбцов отсчитываются от левого верхнего угла заданного диапазона (в данном случае – от ячейки B1).
Для вычисления среднего значения каждой строки используются вложенные циклы For. Эта конструкция рассмотрена в подразделе 1.5.
В операторе d.Cells(i, n + 1).Value = srednee вычисленное среднее значение выводится в ячейку с номером строки i и номером столбца n+1 относительно диапазона d, т.е. в столбец справа от этого диапазона.
Примечание – Если бы в программе отсутствовал оператор Set d = Range("B1:E5"), то для ссылки на диапазон ячеек каждый раз требовалось бы указывать его. Например, для определения количества строк потребовалось бы указать: m = Range("B1:E5").Rows.Count.
Конечно, рассмотренную задачу можно было решить и многими другими способами. Например, можно было воспользоваться ссылкой на ячейки в форме Cells:
Sub primer2_6a_1()
Set d = Range(Cells(1, 2), Cells(5, 5))
… 
Здесь вместо обозначения ячейки B1 использовано обозначение Cells(1, 2), а вместо E5 - Cells(5, 5).
Рассмотрим еще один способ решения рассмотренной задачи: содержимое ячеек вводится в массив, который затем обрабатывается.
Sub primer2_6a_2()
Dim a() As Single, srednie() As Single
Set d = Range("B1:E5")
m = d.Rows.Count
n = d.Columns.Count
ReDim a(1 To m, 1 To n), srednie (1 To m)
For i = 1 To m
For j = 1 To n
a(i, j) = d.Cells(i, j).Value
Next j
Next i
For i = 1 To m
sum = 0
For j = 1 To n
sum = sum + a(i, j)
Next j
srednie(i) = sum / n
Next i
For i = 1 To m
d.Cells(i, n + 1).Value = srednie(i)
Next i
End Sub
Здесь значения ячеек диапазона B1:E5 считываются в двумерный массив a. Затем вычисляются средние значения строк этого массива. Эти средние значения сохраняются в одномерном массиве srednie. Элементы этого массива затем выводятся в ячейки справа от диапазона B1:E5.
б) Вычисления выполняются для произвольного диапазона ячеек, выделенного с помощью мыши.
Программа для решения этой задачи отличается от приведенной в примере 2.6a только тем, что оператор Set d = Range("B1:E5") требуется заменить на Set d = Selection. Здесь переменной d присваивается диапазон ячеек, выделенный с помощью мыши. В операторах m = d.Rows.Count и n = d.Columns.Count переменные m и n получают значения, равные количеству строк и столбцов этого диапазона. Дальнейшие действия выполняются аналогично примерам, рассмотренным выше.
в) Вычисления выполняются для произвольного прямоугольного диапазона ячеек. Известно, что одна из ячеек этого диапазона – B1.
В программе, приведенной в примере 2.6a, требуется заменить оператор Set d = Range("B1:E5") на Set d = Range("B1").CurrentRegion. Здесь переменной d присваивается заполненный данными прямоугольный диапазон ячеек, одна из которых – B1.
г) Вычисления выполняются для произвольного прямоугольного диапазона ячеек, одна из которых выделена с помощью мыши.
В программе, приведенной в примере 2.6a, требуется заменить оператор Set d = Range("B1:E5") на Set d = Selection.CurrentRegion.
Пример 2.7 – На рабочем листе Excel в столбце C, начиная с ячейки C2 (т.е. в ячейках C2, C3 и т.д.), введены фамилии студентов, а в столбцах D, E, F, G – их оценки по четырем предметам. Требуется получить в столбце L (начиная с ячейки L1) список студентов, имеющих средний балл не ниже минимально допустимого. В столбце M рядом с фамилиями студентов должны выводиться их средние баллы. Минимально допустимый средний балл вводится с клавиатуры.
Sub primer2_7()
Dim min_ball As Single
min_ball = InputBox("Введите минимально допустимый средний балл: ")
Set d = Range("C2").CurrentRegion
m = d.Rows.Count
n = d.Columns.Count
k = 0
For i = 1 To m
sum = 0
For j = 2 To n
sum = sum + d.Cells(i, j).Value
Next j
srednee = sum / (n-1)
If srednee >= min_ball Then
k = k + 1
Cells(k, 12).Value = d.Cells(i,1).Value
Cells(k, 13).Value = srednee
End If
Next i
End Sub
Здесь в операторе Set d = Range("C2").CurrentRegion переменной d присваивается диапазон заполненных ячеек, заполненных данными (фамилиями и оценками); одна из этих ячеек – C2. Переменная m получает значение – количество строк диапазона d (в данном примере – количество студентов, для которых введены данные). Переменная n – количество столбцов диапазона d. Если данные введены в соответствии с постановкой задачи, то переменная n должна получить значение 5, так как диапазон данных в этом примере содержит пять столбцов: в столбце C – фамилии студентов, в столбцах D-G – их оценки.
Цикл For i = 1 To m предназначен для перебора строк (каждая строка содержит данные об одном студенте). Для каждого студента вычисляется средний балл. Цикл For j = 2 To n предназначен для перебора оценок студента (т.е. столбцов). Начальное значение переменной j, используемой в качестве номера столбца, равно двум (а не одному), так как оценки, по которым вычисляются среднее, начинаются со второго столбца в диапазоне данных (в первом столбце находятся фамилии, а не оценки). Величина d.Cells(i, j).Value – это значение ячейки с оценкой студента. Следует еще раз обратить вимание, что номера ячеек (значения i и j) отсчитываются в пределах диапазона d, т.е. от ячейки C2.
Если вычисленный средний балл оказывается не ниже заданной минимальной величины (т.е. выполняется условие srednee >= min_ball), то фамилия студента выводится в столбец L (в 12-й столбец рабочего листа), а его средний балл – в столбец M (13-й столбец). Переменная k – номер строки рабочего листа, куда выводится фамилия студента; при выводе каждой фамилии она увеличивается на единицу.
Следует обратить внимание на строку Cells(k,12).Value=d.Cells(i,1).Value. Здесь Cells(k, 12).Value – ячейка, расположенная в k-й строке и 12-м столбце (т.е. столбце L) рабочего листа Excel. Этой ячейке присваивается значение d.Cells(i, 1).Value, т.е. содержимое ячейки, расположенной в i-й строке и первом столбце диапазона d (в этой ячейке находится фамилия студента).
Пример 2.8 – На рабочем листе Excel введены данные о студентах (см. исходные данные для примера 2.7). Требуется удалить данные обо всех студентах, имеющих средний балл ниже минимально допустимого.
Sub primer2_8()
Dim min_ball As Single
min_ball = InputBox("Введите минимально допустимый средний балл: ")
Set d = Range("C2").CurrentRegion
m = d.Rows.Count
n = d.Columns.Count
i = 1
Do While i <= m
sum = 0
For j = 2 To n
sum = sum + d.Cells(i, j).Value
Next j
srednee = sum / (n-1)
If srednee < min_ball Then
For k = i to m-1
For j = 1 to n
d.Cells(k, j).Value = d.Cells(k+1, j).Value
Next j
Next k
For j = 1 to n
d.Cells(m, j).Value = Empty
Next j
m = m-1
Else
i = i+1
End If
Loop
End Sub
В данном случае количество строк в диапазоне с данными (переменная m) не является постоянной величиной, а может изменяться, так как некоторые строки (данные о студентах) будут удаляться. Поэтому для перебора строк использован цикл Do While.
Если для i-го студента (т.е. в i-й строке) вычисленный средний балл оказывается ниже заданной минимальной величины, то данные об этом студенте требуется удалить. Следующая группа операторов смещает все последующие данные, начиная с i+1-й строки, на одну строку выше:
For k = i to m-1
For j = 1 to n
d.Cells(k, j).Value = d.Cells(k+1, j).Value
Next j
Next k
Другими словами, в i-ю строку записываются данные из i+1-й строки, в i+1-ю – из i+2-й строки, …, в m-1-ю – из m-й (т.е. из последней) строки. Затем последняя (m-я) строка очищается: ее ячейкам присваивается значение Empty.
После этого общее количество строк в диапазоне (или, другими словами, номер последней строки) уменьшается на единицу: m = m-1.
Если же для i-го студента средний балл оказался не ниже минимально необходимого, то просто выполняется переход к данным о следующем студенте (к следующей строке диапазона данных): i = i+1.
Примечание – В этом примере показано, как очистить ячейку. Если требуется проверить, пуста ли ячейка, то используется функция IsEmpty. Например, функция IsEmpty(Cells(1,5).Value) возвращает значение True, если ячейка E1 пуста, и значение False, если в этой ячейке есть какая-либо величина.
Пример 2.9 – На рабочем листе Excel введены данные о контрактах на поставку некоторых товаров (по каждому контракту поставляется один товар). Для каждого контракта в столбце A указан его номер, в столбце B – название поставляемого товара, в столбце C – количество товара. Один и тот же товар может поставляться по нескольким контрактам, но цена товара во всех контрактах одинакова. В столбце F перечислены названия товаров, в столбце G – цены на них. Фрагмент исходных данных для задачи приведен на рисунке 2.1. Требуется вычислить и вывести в столбец D стоимость каждого контракта.
[image: ]
Рисунок 2.1 – Фрагмент исходных данных для примера 2.9
Sub primer2_9()
Set d1 = Range("A1").CurrentRegion ‘d1 – диапазон ячеек с данными о контрактах
Set d2 = Range("F1").CurrentRegion ‘d2 - диапазон ячеек с данными о товарах
m1 = d1.Rows.Count ‘ Количество контрактов
m2 = d2.Rows.Count ‘ Количество товаров
For i = 1 To m1          ‘ Перебор всех контрактов
nazv = d1.Cells(i,2).Value ‘ Переменной nazv присваивается название товара, поставляемого
                                           ‘ по контракту
kol = d1.Cells(i,3).Value   ‘ Переменной kol присваивается количество поставляемого товара
For j = 1 To m2                  ‘ Перебор данных о товарах (поиск цены на товар)
If nazv = d2.Cells(j,1).Value then ‘Если товар найден
cena = d2.Cells(j,2).Value             ‘Цена на товар присваивается переменной cena
stoimost = cena*kol                       ‘ Вычисление стоимости контракта
d1.Cells(i,4).Value = stoimost       ‘ Стоимость выводится в i-ю строку, столбец D 
End If
Next j
Next i
End Sub
В данном примере используются два диапазона: d1 – диапазон ячеек с данными о контрактах, d2 – с данными о товарах и ценах на них. Если считать, что на рисунке 2.1 показаны все данные, то диапазон d1 содержит ячейки A1:C7, а d2 – ячейки F1:G4. Переменная i используется в качестве номера строки в диапазоне d1, переменная j – также номер строки, но в диапазоне d2.
2.9	Варианты заданий
Примечание – В этой лабораторной работе рекомендуется выполнить один из вариантов 1-6 и один из вариантов 7-12.
Вариант 1 – На рабочем листе Лист1 в столбце A введены номера контрактов, в столбце B – названия товаров (для каждого контракта – один товар), в столбце C - количество товара. На рабочем листе Лист2 в столбце A введены названия товаров (каждый товар указан один раз), в столбце B - цены товаров. Предполагается, что цена каждого из товаров во всех контрактах одинакова. Программа должна выводить на рабочий лист Лист3 информацию обо всех контрактах на поставку заданного товара: для каждого контракта выводится его номер, количество товара и стоимость. Кроме того, должно выводиться общее количество товара и общая стоимость по всем контрактам на заданный товар. Товар выбирается на рабочем листе Лист2 выделением ячейки с его названием.
Вариант 2 – На рабочем листе Лист1 в столбце A введены названия товаров, в столбце B – цены этих товаров, в столбце C – названия валют, в которых указаны цены (может быть указано несколько товаров, цены которых выражены в одной и той же валюте). На рабочем листе Лист2 в столбце A перечислены названия валют (каждая валюта указана один раз), в столбце B – их курсы в долларах. Программа должна выводить на рабочий лист Лист3 перечень названий товаров и их цены в валюте, выбранной выделением ячейки на рабочем листе Лист2. Указание: если валюта, выбранная на листе Лист2 – не доллар, то для вычисления цены товара в выбранной валюте следует сначала вычислить эту цену в долларах, а затем пересчитать в выбранную валюту.
Вариант 3 – На рабочем листе Лист1 в столбце A введены фамилии работников, в столбце B – номера отделов, где они работают (в одном отделе может быть несколько работников), в столбце C – их зарплаты. На этом же листе в столбце G перечислены номера отделов, в столбце H – коэффициенты повышения зарплаты для работников данного отдела. Для каждого отдела коэффициент повышения зарплаты указывается только один раз, т.е. он одинаков для всех работников отдела. Программа должна вычислять новые зарплаты и выводить их в столбец C взамен старых. Кроме того, на рабочий лист Лист2 должен выводиться перечень отделов с указанием количества работников и суммы их зарплат по каждому отделу.
Вариант 4 – На рабочем листе Лист1 в столбце A введены номера контрактов, в столбце B – названия товаров, продаваемых по этим контрактам (по каждому контракту – один товар), в столбце C – цены, по которым предполагается продавать товары. На рабочем листе Лист2 в столбце A перечислены названия товаров (каждый товар – один раз), в столбце B – предельные минимальные цены на них, в столбце C – предельные максимальные цены (товары запрещается продавать по ценам ниже минимальных или выше максимальных). Для некоторых товаров одна из предельных цен может быть не указана.
Программа должна для каждого контракта, где цена выходит за установленные пределы, заменять ее на соответствующую предельную цену. Кроме того, на рабочем листе Лист3 должен быть получен перечень номеров контрактов, для которых цена была исправлена.
Вариант 5 – На рабочем листе Лист1 в столбце A введены номера контрактов, в столбце B – названия товаров, проданных по этим контрактам (по каждому контракту – один товар), в столбце C – количество товара, проданного по каждому контракту, в столбце D - цены, по которым проданы товары, в столбце E – названия заказчиков (у каждого контракта один заказчик). На рабочем листе Лист2 в столбце A перечислены названия заказчиков. Программа должна выводить на рабочий лист Лист3 следующую информацию о контрактах выбранного заказчика: количество контрактов; общая стоимость контрактов; номер контракта максимальной стоимости. Заказчик выбирается на рабочем листе Лист2 выделением ячейки с его названием.
Вариант 6 – На рабочем листе Лист1 в столбце A введены номера контрактов, в столбце B – названия товаров, проданных по этим контрактам (по каждому контракту – один товар), в столбце C – количество товара, проданного по каждому контракту, в столбце D - цены, по которым проданы товары. На рабочем листе Лист2 в столбце A перечислены названия товаров (каждый товар – один раз), в столбце B – ставки налогов по контрактам на эти товары. Например, если в ячейке A1 указано название товара – компьютер, а в ячейке B1 – ставка 12%, это означает, что с каждого контракта на поставку компьютеров выплачивается налог в размере 12% от его полной стоимости.
Программа должна для каждого контракта вычислять выплачиваемый за него налог и выводить его в столбце E. Кроме того, на рабочем листе Лист3 должен быть получен перечень всех товаров с указанием суммы налогов по всем контрактам на данный товар.
Вариант 7 – Произвольная прямоугольная область рабочего листа Лист1 заполнена числами. Программа должна выводить на рабочий лист Лист2 в столбец A номера строк заполненной области, содержащих заданное число, а в столбец B - номер первого вхождения заданного числа. Например, если задано число 5, то в столбец A должны выводиться номера строк, содержащих хотя бы одно число 5, а в столбец B – номера столбцов, где находится первое число 5 в каждой из этих строк. Заданное число вводится с клавиатуры.
Вариант 8 – Произвольная прямоугольная область рабочего листа Лист1 заполнена числами. Программа должна определять в этой области первый столбец, сумма которого превосходит заданную величину (эта величина вводится с клавиатуры), и копировать этот столбец на рабочий лист Лист2. Из исходной области выбранный столбец должен удаляться, а столбцы, расположенные после него – смещаться на один столбец влево.
Вариант 9 – На рабочем листе имеются два столбца чисел, причем известно, что сумма первого из этих столбцов больше, чем сумма второго. Программа должна переносить числа из конца первого столбцов в конец второго до тех пор, пока сумма второго столбца не превысит сумму первого.
Вариант 10 – На рабочем листе выделен некоторый прямоугольный диапазон. Программа должна удалять из него все пустые строки, а также все строки, заполненные нулями.
Вариант 11 – На рабочем листе имеется заполненная числами прямоугольная область из m строк и n столбцов. Каждая строка представляет собой координаты некоторой точки в n-мерном пространстве. 
На этом же рабочем листе в другом месте расположена строка из n чисел – координаты еще одной точки в n-мерном пространстве. Программа должна найти для нее первую из m точек, расположенную на расстоянии меньше заданного (заданное расстояние вводится с клавиатуры).
Вариант 12 – На рабочем листе имеется заполненная числами прямоугольная область. Из каждой строки этой области программа должна копировать элементы (начиная с максимального, затем – второго по величине, и т.д.) в другую область рабочего листа до тех пор, пока сумма элементов, скопированных из строки, не превысит некоторую заданную величину, вводимую с клавиатуры. Если это окажется невозможным (сумма всей строки окажется не больше заданной величины), то должно выводиться сообщение об ошибке. При выполнении этой задачи исходная область, заполненная числами, не должна изменяться.
Контрольные вопросы.

1. Перечислите типы констант  в VBA?
2. Как объявляются переменные в VBA?
3. Назовите типы переменных по умолчанию в VBA?
4. Назовите  основные встроенные математические функции в VBA.
5. Назовите приоритет выполнения математических операций в арифметических выражениях в VBA.
6. Перечислите простейшие средства ввода-вывода в VBA.
7. Поясните суть встроенных диалоговых окон VBA для ввода и вывода данных в программе.
8. Из каких основных компонентов состоит интерфейс VBA?
Оформление отчета
Отчет по лабораторной работе должен содержать:
1. Титульный лист.
2.  Цели работы.
3.  Краткие теоретические сведения в виде ответов на контрольные вопросы. 
4.  Результаты лабораторной работы (графики, таблицы, значения параметров измеренных в процессе работы).
5.  Необходимые расчеты (если они оговариваются в практической части лабораторной работы).
6.  Выводы по работе (что получено в процессе работы, почему получены данные результаты и т. д.).
7.  Отчет для защиты представляется преподавателю в отпечатанном или электронном виде.

